unittest
{
	assert( [ "3 2" ].parse.expand.solve == 10 );
	assert( [ "200000 1000000000" ].parse.expand.solve == 607923868 );
	assert( [ "15 6" ].parse.expand.solve == 22583772 );
}



import std.algorithm;
import std.conv;
import std.range;
import std.stdio;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	auto nk = input.front.split.to!( long[] );
	return tuple( nk[ 0 ], nk[ 1 ] );
}

auto solve( long n, long k )
{
	// 移動を 0,1,2 -> 1,0,2 のようにすれば k-1 のときと同じパターンを作れる
	// つまり、0 ~ k のパターンを全部考慮する必要がある
	
	// 0人部屋の数を決め打ちする
	auto ans = ModWorld!()( 0 );
	foreach( i; 0L .. min( n, k + 1 ) )
	{
		// 0人部屋のパターン数を計算
		auto a = n.combination( i );
		
		// 0人部屋に居た人の移動先のパターン数を計算
		auto p = n - i - 1;
		auto b = ( p + i ).combination( p );
		
		ans += a + b;
	}
	return ans;
}

// モジュラ整数型（n mod M）
struct ModWorld( long M = 10L ^^ 9 + 7 )
{
	long n;
	
	// 演算子オーバーロード
	void opUnary( string OP )() if( OP == "++" ){ this.n = ( this.n + 1 ) % M; }
	void opUnary( string OP )() if( OP == "--" ){ this.n = ( this.n + M - 1 ) % M; }
	void opOpAssign( string OP )( ModWorld r ) if( OP == "+"  ){ this.n = ( this.n + r.n ) % M; }
	void opOpAssign( string OP )( ModWorld r ) if( OP == "-"  ){ this.n = ( this.n + M - r.n ) % M; }
	void opOpAssign( string OP )( ModWorld r ) if( OP == "*"  ){ this.n = ( this.n * r.n ) % M; }
	void opOpAssign( string OP )( ModWorld r ) if( OP == "/"  ){ this *= r ^^ ModWorld( M - 2 ); }
	void opOpAssign( string OP )( ModWorld r ) if( OP == "^^" )
	{
		auto a = this.n;
		this.n = 1L;
		for( auto i = r.n; 0 < i; i >>= 1 )
		{
			if( i & 1 ) this.n = this.n * a % M;
			a = a ^^ 2 % M;
		}
	}
	auto opBinary( string OP )( ModWorld r )
	{
		auto t = this;
		t.opOpAssign!OP( r );
		return t;
	}
	
	// long型対応
	void opAssign( long r ){ this.n = r; }
	void opOpAssign( string OP )( long r ){ this.opOpAssign!OP( ModWorld( r ) ); }
	auto opBinary( string OP )( long r ){ return this.opBinary!OP( ModWorld( r ) ); }
	auto opBinaryRight( string OP )( long l ){ return ModWorld( l ).opBinary!OP( this ); }
	auto opEquals( long r ){ return this.n == r; }
	
	// format用出力
	string toString()
	{
		return this.n.to!string;
	}
}

// 階乗
auto factorial( long M = 10L ^^ 9 + 7 )( long x )
{
	static memo = [ ModWorld!M( 1L ) ];
	if( memo.length <= x )
	{
		auto i = memo.length.signed;
		memo.length = x + 1;
		for( ; i <= x; i++ ) memo[ i ] = memo[ i - 1 ] * i;
	}
	return memo[ x ];
}

// 組み合わせ数
auto combination( long M = 10L ^^ 9 + 7 )( long x, long y )
{
	if( x < y ) return ModWorld!M( 0L );
	else        return x.factorial!M / ( y.factorial!M * ( x - y ).factorial!M );
}
